



Scripting Apache OpenOffice

Introductory
Nutshell Programs
(Writer, Calc, Impress)

Rony G. Flatscher

FOSDEM '13

Overview

- Overview of AOO
 - Bird eye's view of AOO's architecture
- Scripting AOO
- Nutshell examples
 - “swriter” (word processor), “scalc” (spreadsheet), “sdraw” (drawing), “sypress” (presentation)
- Roundup
- Links



Bird Eye's View, 1

- Set of *services* that may contain *interfaces* with *attributes*, other *services*, *structs* and *properties*
- All common functionality of all types of documents is extracted and organized as a set of *interfaces* that define *methods* and possibly *attributes*
 - E.g. loading, saving, printing documents, ...
- *Services* are created and get managed by *service managers*



Bird Eye's View, 2

- Client-/Server-Architecture
 - Communication via TCP/IP
 - Employing distributable components (“UNO”)
 - Server can run on any computer in the world!
 - Operating systems of the server and the client are irrelevant for the purpose of communication!
 - Client may run on the same machine as the server
 - Default installation and configuration




Bird Eye's View, 3


- “UNO”
 - **U**niversal **N**etwork **O**bjects
 - Distributable, interconnected infrastructure
 - All functionality is organized in the form of classes (“UNO classes”)
 - UNO classes (types) get defined in an IDL (Interface Description Language)
- “urp”
 - **U**niversal **r**emote **p**rotocol
 - CORBA-like




Bird Eye's View, 4



Bird Eye's View, 5



Bird Eye's View, 6



Bird Eye's View, 7

- “Service Managers” (a.k.a. “factories”)
 - Supplied by servers
 - Also cf. `XComponentContext.getServiceManager()`
 - Can be used to request/create *services*
 - Returned *service* allows access to a part of the "office" functionality, e.g.
 - `com.sun.star.frame.Desktop`
 - `com.sun.star.configuration.ConfigurationProvider`
 - `com.sun.star.sdb.DatabaseContext`



Bird Eye's View, 8




Illustration 2.1: Service manager



Bird Eye's View, 9

- “Services”
 - Can be comprehensive
 - May contain
 - “Interfaces” (group of *methods* and *attributes*)
 - Other “Services”
 - “properties” ([com.sun.star.beans.PropertyValue](#))
 - Depending on the desired task you need to query (request) the appropriate interface, e.g.
 - [com.sun.star.view.XPrintable](#)
 - [com.sun.star.frame.XStorable](#)
 - [com.sun.star.text.XTextDocument](#)



Bird Eye's View, 10

- An example
 - Two services with seven interfaces
 - "OfficeDocument"
 - Four interfaces
 - "TextDocument"
 - Three interfaces




Illustration 2.3: Text Document



Scripting AOO

Programming Languages

- Programming languages
 - C++ (*queryInterface*)
 - Java (*queryInterface*)
 - Basic (implicit *queryInterface*)
 - Python (implicit *queryInterface*)
- Java-based scripting framework
 - BeanShell (*queryInterface*)
 - JavaScript (*queryInterface*)
 - ooRexx (*queryInterface*)
 - ...



Scripting AOO Documentation

- *Extremely* important
 - Wealth of services and interfaces
 - Created in pure German ;) engineering style
 - To miss the the forest for the trees!
- AOO API documentation
 - <http://www.openoffice.org/api/>
 - Developer's guide, API wiki, UNO wiki, extensions, examples, tutorials
 - <http://www.openoffice.org/api/docs/common/ref/com/sun/star/module-ix.html>
 - Extensive, HTML-linked API reference
 - Use its Index to locate services, interfaces, etc.



Module star - Mozilla Firefox

File Edit View History Bookmarks Tools Help OO.o

Module star Global Index A Global Index X

www.openoffice.org/api/docs/common/ref/com/sun/star

Disable Cookies CSS Forms Images Information Miscellaneous Outline Resize Tools View Source Options

Apache OpenOffice™ The Free and Open Productivity Suite

OpenOffice graduates from the Apache Incubator!

home » api » docs » common » ref » com » sun » star | Product | Download | Support | Extend | Develop | Focus Areas | Native Language

Content for Apache OpenOffice version 3.4.

Overview Module Use Devguide **Index**

NESTED MODULES SERVICES SINGLETONS INTERFACES STRUCTS EXCEPTIONS ENUMS TYPEDEFS CONSTANT GROUPS

:: com :: sun ::

module star

Nested Modules

accessibility	UNO Accessibility API
animations	
auth	security and authenticates interfaces
awt	Java AWT-like user interface toolkit interface specifications for UNO.
beans	Java beans-like property access and introspection.
bridge	Interfaces for building bridges to other component models.
chart	Charting diagram interfaces.
chart2	New implementation of Charting diagram interfaces. This module contains only a rather small public API. In addition there is a private API in the chart2 project.

DEVELOPER'S GUIDE
Content Table
IDL reference

API
Module structure

SDK
Examples
Java UNO Reference
C++ UNO Reference
Download

TIPS 'N' TRICKS
FAQ
Internal OO Spots
External Resources

MISCELLANEOUS
Developer Projects
Mailing List Rules

Adblock Plus

Wien, Austria (Vienna): 9°C | Fri: 10°C | Sat: 8°C

Global Index A - Mozilla Firefox

File Edit View History Bookmarks Tools Help OO.o

Module star Global Index A Global Index X

Back Forward www.openoffice.org/api/docs/common/ref/index-files/inc Liveshare Reload Stop Google Home

Disable Cookies CSS Forms Images Information Miscellaneous Outline Resize Tools View Source Options

Apache OpenOffice™ The Free and Open Productivity Suite

OpenOffice graduates from the Apache Incubator!

home » api » docs » common » ref » index-files | Product | Download | Support | Extend | Develop | Focus Areas | Native Language

DEVELOPER'S GUIDE
Content Table
IDL reference

API
Module structure

SDK
Examples
Java UNO Reference
C++ UNO Reference
Download

TIPS 'N' TRICKS
FAQ
Internal OO Spots
External Resources

MISCELLANEOUS
Developer Projects
Mailing List Rules

Content for Apache OpenOffice version 3.4.

Overview Module Use Devguide Index

Global Index A

ABCDEFGHIJKLMNOPQRSTUVWXYZ

A - constant in constants group ::com::sun::star::awt: [.Key](#)
aArgs - field in struct ::com::sun::star::frame: [.DispatchStatement](#)
abbreviateString() - function in interface ::com::sun::star::util: [.XStringAbbreviation](#)
ABBREVIATION - constant in constants group ::com::sun::star::linguistic2: [.ConversionPropertyType](#)
AbbrevName - field in struct ::com::sun::star::i18n: [.CalendarItem](#)
aBitmapMode - field in struct ::com::sun::star::chart2: [.FillBitmap](#)
ABORT - value in enum ::com::sun::star::ucb: [.IOErrorCode](#)
abort() - function in interface ::com::sun::star::ucb: [.XCommandProcessor](#)
Aborted - property in service ::com::sun::star::document: [.MediaDescriptor](#)
aborted() - function in interface ::com::sun::star::sheet: [.XRangeSelectionListener](#)
abortRangeSelection() - function in interface ::com::sun::star::sheet: [.XRangeSelection](#)
ABOVE - constant in constants group ::com::sun::star::awt: [.FontEmphasisMark](#)
AboveCenter - constant in constants group ::com::sun::star::awt: [.ImagePosition](#)
AboveLeft - constant in constants group ::com::sun::star::awt: [.ImagePosition](#)

Adblock Plus x

Wien, Austria (Vienna): 9°C Fri: 10°C Sat: 8°C



Global Index X - Mozilla Firefox

File Edit View History Bookmarks Tools Help OO.o

Module star Global Index A Global Index X

Back Forward www.openoffice.org/api/docs/common/ref/index-files/inc Google Home

Disable Cookies CSS Forms Images Information Miscellaneous Outline Resize Tools View Source Options

Apache OpenOffice™ The Free and Open Productivity Suite

OpenOffice graduates from the Apache Incubator!

home » api » docs » common » ref » index-files | Product | Download | Support | Extend | Develop | Focus Areas | Native Language

DEVELOPER'S GUIDE

- Content Table
- IDL reference

API

- Module structure

SDK

- Examples
- Java UNO Reference
- C++ UNO Reference
- Download

TIPS 'N' TRICKS

- FAQ
- Internal OO Spots
- External Resources

MISCELLANEOUS

- Developer Projects
- Mailing List Rules

Content for Apache OpenOffice version 3.4.

Overview Module Use Devguide Index

Global Index X

A B C D E F G H I J K L M N O P Q R S T U V W X Y Z _

- X** - field in struct ::com::sun::star::awt:: [Point](#)
- X** - constant in constants group ::com::sun::star::awt:: [PosSize](#)
- X** - field in struct ::com::sun::star::awt:: [Rectangle](#)
- X** - field in struct ::com::sun::star::geometry:: [IntegerPoint2D](#)
- X** - field in struct ::com::sun::star::awt:: [MouseEvent](#)
- X** - field in struct ::com::sun::star::geometry:: [RealPoint2D](#)
- X** - field in struct ::com::sun::star::awt:: [WindowEvent](#)
- X** - constant in constants group ::com::sun::star::awt:: [Key](#)
- X** - constant in constants group ::com::sun::star::awt:: [FontStrikeout](#)
- XAbortChannel** - interface ::com::sun::star::task:: [XAbortChannel](#)
- XAbstractView** - interface ::com::sun::star::xml::dom::views:: [XAbstractView](#)
- XAcceleratorConfiguration** - interface ::com::sun::star::ui:: [XAcceleratorConfiguration](#)
- XAcceptor** - interface ::com::sun::star::connection:: [XAcceptor](#)
- XAccessControlContext** - interface ::com::sun::star::security:: [XAccessControlContext](#)

Adblock Plus x

Wien, Austria (Vienna): 9°C Fri: 10°C Sat: 8°C

Scripting AOO

Querying an Interface

- *queryInterface()* examples
 - *sDispatchHelper*, a service of type *com.sun.star.frame.DispatchHelper*

- *queryInterface()* in Java

```
import com.sun.star.frame.XDispatchHelper;  
// ...  
XDispatchHelper xDispatchHelper=(XDispatchHelper)  
    UnoRuntime.queryInterface(XDispatchHelper.class, sDispatchHelper);
```

- *queryInterface()* in JavaScript

```
importClass(Packages.com.sun.star.frame.XDispatchHelper);  
// ...  
xDispatchHelper = UnoRuntime.queryInterface(XDispatchHelper, sDispatchHelper);
```

- *queryInterface()* in ooRexx

```
xDispatchHelper=sDispatchHelper~com.sun.star.frame.XDispatchHelper  
-- or simpler:  
xDispatchHelper=sDispatchHelper~XDispatchHelper
```



Scripting AOO

- Two kinds of scripting (programming)
 - **Stand-alone**
 - Need to bootstrap OpenOffice in order to initialize the AOO environment to interact with
 - Full control about addressing different AOO servers, if needed
 - Dispatched by AOO (“macro”)
 - AOO supplies a script context that allows access to the initialized AOO environment (*getDesktop*, *getComponentContext*) and to the document (*getDocument*) for which the dispatch occurred



Scripting AOO Bootstrapping in Java

```
// import ...
XComponentContext xLocalContext =
com.sun.star.comp.helper.Bootstrap.createInitialComponentContext(null);
// initial serviceManager
XMultiComponentFactory xLocalServiceManager = xLocalContext.getServiceManager();
// create a URL resolver
Object urlResolver = xLocalServiceManager.createInstanceWithContext(
    "com.sun.star.bridge.UnoUrlResolver", xLocalContext);
// query for the XUnoUrlResolver interface
XUnoUrlResolver xUrlResolver = (XUnoUrlResolver)
UnoRuntime.queryInterface(XUnoUrlResolver.class, urlResolver);
// Import the object
Object rInitialObject = xUrlResolver.resolve(
    "uno:socket,host=localhost,port=8100;urp;StarOffice.ServiceManager");
// test whether we got a reference to the remote ServiceManager
if (null != rInitialObject) {
    System.out.println("initial object successfully retrieved");
} else {
    System.out.println("given initial-object name unknown at server side");
}

... cut ...
```



Scripting AOO

Bootstrapping in ooRexx

```
url="uno:socket,host=localhost,port=8100;urp;StarOffice.ServiceManager"
rInitialObject=uno.connect(url)

if rInitialObject<>.nil then
  say "initial object successfully retrieved"
else
  say "given initial-object name unknown at server side"
-- ... cut ...

::requires UNO.CLS -- get UNO support
```



Scripting AOO

Creating/Loading Documents

scal
swriter
simpres
sdraw

```
xDesktop=uno.createDesktop() -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader -- get XComponentLoader interface

uri="private:factory/swriter" -- new swriter document
doc=xcl~loadComponentFromURL(uri, "_blank", 0, .uno~noProps)

-- ... now do whatever you want or need to do ...

::requires UNO.CLS -- get UNO support
```

"file:///c:/docs/aFile.odt"
"http://www.RexxLA.org/aFile.ods"
"ftp://www.OpenOffice.org/aFile.odp"



Nutshell examples

Word Processor (“swriter”), 1

– 3 Services

GenericTextDocument (com.sun.star.text.*GenericTextDocument*),
OfficeDocument (com.sun.star.document.*OfficeDocument*), TextDocument
(com.sun.star.text.*TextDocument*)

– 35 Interfaces (unqualified)

XBookmarksSupplier, XChapterNumberingSupplier,
XDocumentEventBroadcaster, XDocumentIndexesSupplier,
XDocumentInfoSupplier, XDocumentPropertiesSupplier, XEmbeddedScripts,
XEndnotesSupplier, XEventBroadcaster, XEventsSupplier, XFootnotesSupplier,
XLineNumberingSupplier, XModel, XModifiable, XMultiServiceFactory,
XNumberFormatsSupplier, XPagePrintable, XPrintJobBroadcaster, XPrintable,
XPropertySet, XReferenceMarksSupplier, XRefreshable, XReplaceable,
XSearchable, XStorable, XStyleFamiliesSupplier, [XTextDocument](#),
XTextEmbeddedObjectsSupplier, XTextFieldsSupplier, XTextFramesSupplier,
XTextGraphicObjectsSupplier, XTextSectionsSupplier, XTextTablesSupplier,
XUndoManagerSupplier, XViewDataSupplier



Nutshell examples

Word Processor (“swriter”), 2

- 37 Properties

ApplyFormDesignMode, ApplyWorkaroundForB6375613,
AutomaticControlFocus, BasicLibraries, BuildId, CharFontCharSet,
CharFontCharSetAsian, CharFontCharSetComplex, CharFontFamily,
CharFontFamilyAsian, CharFontFamilyComplex, CharFontName,
CharFontNameAsian, CharFontNameComplex, CharFontPitch,
CharFontPitchAsian, CharFontPitchComplex, CharFontStyleName,
CharFontStyleNameAsian, CharFontStyleNameComplex, CharLocale,
CharacterCount, DialogLibraries, ForbiddenCharacters, HasValidSignatures,
HideFieldTips, IndexAutoMarkFileURL, LockUpdates, ParagraphCount,
RecordChanges, RedlineDisplayType, RedlineProtectionKey, RuntimeUID,
ShowChanges, TwoDigitYear, WordCount, WordSeparator



Nutshell examples

Word Processor (“swriter”), 3

- Interface `com.sun.star.text.XTextDocument`
 - Get access to the text object representing the text of the entire document using `getText()`
 - Returns `XText`, which is derived from `XSimpleText`, which is derived from `XRangeText`, hence the methods of all three interfaces are available!
- Concept of “cursors”, e.g.
 - Paragraphs, Sentences, Words, Characters
- Possible to also insert tables, fields, pictures, drawings, ...



Nutshell examples

Word Processor, Example 1/1

- Example 1
 - Create a word processor document
 - Add text “Hello, FOSDEM 2013!”
 - Closing the word processor document manually will cause the “Save”-dialog to appear



Nutshell examples

Word Processor, Example 1/2

```
xDesktop=uno.createDesktop()      -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader     -- get XComponentLoader interface

uri="private:factory/swriter"     -- new swriter document
doc=xcl~loadComponentFromURL(uri,"_blank",0,.uno~noProps)


xText=doc~XTextDocument~getText  -- get text object
xText~setString("Hello, FOSDEM 2013!")

::requires UNO.CLS               -- get UNO support
```



Nutshell examples

Word Processor, Example 1/3



Nutshell examples

Word Processor, Example 2/1

- Example 2
 - Create a word processor document
 - Add text “**Hello, FOSDEM 2013!**”
 - Change state of document to “unmodified”
 - Leftover document can be closed without a save dialog
 - Using interface `com.sun.star.util.XModifiable`
 - Sleep five seconds, then close document
 - Using interface `com.sun.star.util.XCloseable`



Nutshell examples

Word Processor, Example 2/2

```
xDesktop=uno.createDesktop()      -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader     -- get XComponentLoader interface

uri="private:factory/swriter"     -- new swriter document
doc=xcl~loadComponentFromURL(uri,"_blank",0,.uno~noProps)

xText=doc~XTextDocument~getText  -- get text object
xText~setString("Hello, FOSDEM 2013!")

doc~XModifiable~setModified(.false) -- set document to unmodified
call SysSleep 5                  -- sleep 5 seconds
doc~XCloseable~close(.false)     -- close document (window)

::requires UNO.CLS               -- get UNO support
```



Nutshell examples

Word Processor, Example 3/1

- Example 3
 - Create a word processor document
 - Add text “**Hello, FOSDEM 2013!**”
 - Access and show property **CharacterCount**
 - Change state of document to “unmodified”
 - Leftover document can be closed without a save dialog
 - Using interface `com.sun.star.util.XModifiable`
 - Sleep five seconds, then close document
 - Using interface `com.sun.star.util.XCloseable`



Nutshell examples

Word Processor, Example 3/2

```
xDesktop=uno.createDesktop()           -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader          -- get XComponentLoader interface

uri="private:factory/swriter"          -- new swriter document
doc=xcl~loadComponentFromURL(uri,"_blank",0,.uno~noProps)

xText=doc~XTextDocument~getText       -- get text object
xText~setString("Hello, FOSDEM 2013!")

xprops=doc~XPropertySet               -- get access to the properties
say "character count:" xprops~getPropertyValue("CharacterCount")

doc~XModifiable~setModified(.false)  -- set document to unmodified
call SysSleep 5                       -- sleep 5 seconds
doc~XCloseable~close(.false)          -- close document (window)

::requires UNO.CLS                     -- get UNO support
```

```
E:\rony\Vortraege\2013\FOSDEM\code>rexx swriter3.rxo
character count: 29
```



Nutshell examples

Word Processor, Example 4/1

- Example 4
 - Create a word processor document
 - Add text “Hello, FOSDEM 2013!”
 - Replace “FOSDEM” with “FOSDEM Conference”
 - Change the color to red
 - Change the font name to “DejaVus Sans Mono”



Nutshell examples

Word Processor, Example 4/2

```
xDesktop=uno.createDesktop()           -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader          -- get XComponentLoader interface

uri="private:factory/swriter"          -- new swriter document
doc=xcl~loadComponentFromURL(uri,"_blank",0,.uno~noProps)

xText=doc~XTextDocument~getText       -- get text object
xText~setString("Hello, FOSDEM 2013!")

    -- change second word
xTextCursor=xText~createTextCursor    -- character based cursor
xTextCursor~gotoStart(.false)         -- make sure we are at start

xWordCursor=xTextCursor~XWordCursor   -- get the XWordCursor interface
xWordCursor~gotoNextWord(.false)      -- XTextRange represents first word
xWordCursor~gotoNextWord(.true)       -- select second word, includes blank!
xWordCursor~setString("Apache Conference ") -- note trailing blank

    -- change color
red=box("int", "FF 00 00"x ~c2d)        -- color red (RGB color) as integer
xWordCursor~XPropertySet~setProperty("CharColor", red)


    -- change font
fontName="DejaVu Sans Mono"
xWordCursor~XPropertySet~setProperty("CharFontName", fontName)
say ppd(xWordCursor~uno.getDefinition)

::requires UNO.CLS -- get UNO support
```



Nutshell examples

Word Processor, Example 4/3



Nutshell examples

Word Processor, Example 5/1

- Example 5
 - Create a word processor document
 - Add text “**Hello, FOSDEM 2013!**”
 - Demonstrate creating and styling paragraphs
 - Get access to the paragraph properties
 - Access `com.sun.star.text.ControlCharacter` constants
 - Access to `com.sun.star.style.ParagraphAdjust` enums
 - Demonstrate adjusting paragraphs to “**right**”, “**center**”, “**block**”, “**left**” using a string that contains the adjustment verb



Nutshell examples

Word Processor, Example 5/2

```
xDesktop=uno.createDesktop()           -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader          -- get XComponentLoader interface

uri="private:factory/swriter"          -- new swriter document
doc=xcl~loadComponentFromURL(uri,"_blank",0,.uno~noProps)

xText=doc~XTextDocument~getText        -- get text object
xText~setString("Hello, FOSDEM 2013!")

xTextCursor=xText~createTextCursor     -- create the character based cursor
  -- make paragraph's properties accessible:
xParaProps=xTextCursor~XParagraphCursor~XPropertySet

ctlChars=.uno_constants~new("com.sun.star.text.ControlCharacter") -- UNO_CONSTANT
paraBreak=ctlChars~paragraph_break    -- get paragraph break constant

paraAdj =.uno_enum~new("com.sun.star.style.ParagraphAdjust")      -- UNO_ENUM


arr=.array~of("right", "center", "block", "left") -- adjustments
do adj over arr -- iterate over adjustments, create string, adjust
  xTextCursor~gotoEnd(.false) -- position at end
  xText~insertControlCharacter(xTextCursor, paraBreak, .false)
  string=("This paragraph will be" adj"-adjusted. ")~copies(8)
  xText~insertString(xTextCursor, string, .true)
  xParaProps~setProperty("ParaAdjust", paraAdj~send(adj))
end

::requires UNO.CLS -- get UNO support
```



Nutshell examples

Word Processor, Example 5/3



Untitled 1 - OpenOffice.org Writer

File Edit View Insert Format Table Tools Window Help

Default Times New Roman 12 B I U

1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19

1
1
2
3
4
5
6
7
8
9

Hello, FOSDEM 2013!

This paragraph will be right-adjusted. This paragraph will be right-adjusted. This paragraph will be right-adjusted. This paragraph will be right-adjusted. This paragraph will be right-adjusted. This paragraph will be right-adjusted. This paragraph will be right-adjusted. This paragraph will be right-adjusted.

This paragraph will be center-adjusted. This paragraph will be center-adjusted. This paragraph will be center-adjusted. This paragraph will be center-adjusted. This paragraph will be center-adjusted. This paragraph will be center-adjusted. This paragraph will be center-adjusted.

This paragraph will be block-adjusted. This paragraph will be block-adjusted. This paragraph will be block-adjusted. This paragraph will be block-adjusted. This paragraph will be block-adjusted. This paragraph will be block-adjusted. This paragraph will be block-adjusted.

This paragraph will be left-adjusted. This paragraph will be left-adjusted. This paragraph will be left-adjusted. This paragraph will be left-adjusted. This paragraph will be left-adjusted. This paragraph will be left-adjusted. This paragraph will be left-adjusted.

Page 1 / 1 Default German (Austria) INSRT STD * 100%

Nutshell examples

Spreadsheet (“scalc”), 1

- 3 Services

OfficeDocument (com.sun.star.document.*OfficeDocument*),
SpreadsheetDocument (com.sun.star.sheet.*SpreadsheetDocument*),
SpreadsheetDocumentSettings
(com.sun.star.sheet.*SpreadsheetDocumentSettings*)

- 26 Interfaces (unqualified)

XActionLockable, XCalculatable, XConsolidatable, XDocumentAuditing,
XDocumentEventBroadcaster, XDocumentInfoSupplier,
XDocumentPropertiesSupplier, XDrawPagesSupplier, XEmbeddedScripts,
XEventBroadcaster, XEventsSupplier, XGoalSeek, XLinkTargetSupplier, XModel,
XModifiable, XMultiServiceFactory, XNumberFormatsSupplier,
XPrintJobBroadcaster, XPrintable, XPropertySet, XProtectable,
[XSpreadsheetDocument](#), XStorable, XStyleFamiliesSupplier,
XUndoManagerSupplier, XViewDataSupplier



Nutshell examples

Spreadsheet (“scal”), 2

- 40 Properties

ApplyFormDesignMode, AreaLinks, AutomaticControlFocus, BasicLibraries, BuildId, CalcAsShown, CharLocale, CharLocaleAsian, CharLocaleComplex, CodeName, ColumnLabelRanges, DDELinks, DatabaseRanges, DefaultTabStop, DialogLibraries, ExternalDocLinks, ForbiddenCharacters, HasDrawPages, HasValidSignatures, IgnoreCase, IsAdjustHeightEnabled, IsChangeReadOnlyEnabled, IsExecuteLinkEnabled, IsIterationEnabled, IsLoaded, IsUndoEnabled, IterationCount, IterationEpsilon, LookUpLabels, MatchWholeCell, NamedRanges, NullDate, ReferenceDevice, RegularExpressions, RowLabelRanges, RuntimeUID, SheetLinks, SpellOnline, StandardDecimals, VBAGlobalConstantName



Nutshell examples

Spreadsheet (“scalc”), 3

- Interface `com.sun.star.sheet.XSpreadsheetDocument`
 - Get name access to the collection of `XSpreadsheets`
 - Numeric (0-based) access with `XIndexAccess`
- Concept of “table” consisting of a collection of *rows*, which each have *columns*
 - `XCellRange` (a tabular area of a spreadsheet)
 - Origin “0,0” represents upper left-hand corner
 - Offsets relative to upper left-hand corner



Nutshell examples

Spreadsheet (“scalc”), 4

- Addressing a cell
 - Numerically (0-based) representing offsets from origin
 - e.g. “0,1” (first column, second row)
 - `getCellByPosition(columnOffset,rowOffset)` returns a `XCell`
 - By name
 - a named range, or
 - column: a name, row: a 1-based number), e.g. “A2”
 - `getCellRangeByName(Name)` returns a `XCellRange`, then
 - `getCellByPosition(0,0)` returns a `XCell`
 - Possible to also insert charts, drawings, ...



Nutshell examples

Spreadsheet, Example 1/1

- Example 1
 - Create a spreadsheet document
 - Add text “**Hello, FOSDEM 2013!**” to A1
 - Demonstrate how to store a document



Nutshell examples

Spreadsheet, Example 1/2

```
xDesktop=uno.createDesktop()           -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader          -- get XComponentLoader interface

uri="private:factory/scalc"            -- new scalc document
doc=xcl~loadComponentFromURL(uri,"_blank",0,.uno~noProps)

xSheets=doc~XSpreadSheetDocument~getSheets~XIndexAccess
xSheet =xSheets~getByIndex(0)~XSpreadSheet  -- get first spreadsheet
                                           -- add entry to "A1"
xSheet~getCellByPosition(0,0)~setFormula("Hello, FOSDEM 2013!")

storeURL=directory()"/scalc1.ods"      -- save document in local directory
storeURL=uno.convertToUrl(storeURL)    -- change path to URL-style
doc~XStorable~storeAsURL(storeURL,.UNO~noProps) -- save document

doc~XCloseable~close(.false)          -- close document (window)

::requires UNO.CLS                     -- get UNO support
```



Nutshell examples

Spreadsheet, Example 1/3

The screenshot shows the OpenOffice Calc application window titled "scalc1.ods - OpenOffice.org Calc". The menu bar includes File, Edit, View, Insert, Format, Tools, Data, Window, and Help. The toolbar contains various icons for file operations, editing, and formatting. The font settings are set to Arial, size 10, with bold, italic, and underline options. The active cell is A1, and the formula bar shows the text "Hello, FOSDEM 2013!". The spreadsheet grid shows columns A through H and rows 1 through 23. The text "Hello, FOSDEM 2013!" is entered in cell A1. The status bar at the bottom shows "Sheet 1 / 3", "Default", "STD", "Sum=0", and a zoom level of 100%.

	A	B	C	D	E	F	G	H
1	Hello, FOSDEM 2013!							
2								
3								
4								
5								
6								
7								
8								
9								
10								
11								
12								
13								
14								
15								
16								
17								
18								
19								
20								
21								
22								
23								

Nutshell examples

Spreadsheet, Example 2/1

- Example 2
 - Create a spreadsheet document
 - Add text “**Hello, FOSDEM 2013!**” to A1
 - Demonstrate how to change the height of table rows



Nutshell examples

Spreadsheet, Example 2/2

```
xDesktop=uno.createDesktop()           -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader          -- get XComponentLoader interface

uri="private:factory/scalc"            -- new scalc document
doc=xcl~loadComponentFromURL(uri,"_blank",0,.uno~noProps)

xSheets=doc~XSpreadSheetDocument~getSheets~XIndexAccess
xSheet =xSheets~getByIndex(0)~XSpreadSheet -- get first spreadsheet
                                           -- add entry to "A1"
xSheet~getCellByPosition(0,0)~setFormula("Hello, FOSDEM 2013!")

xRows=xSheet~XColumnRowRange~getRows-- get XTableRows

do i=1 to 5                             -- 0-based, hence lines # 2 through # 6
  xRow=xRows~getByIndex(i)               -- fetch XRow
  props=xRow~XPropertySet                -- get access to its properties
  oldHeight=props~getPropertyValue("Height") -- get current value
  newHeight=oldHeight+i*250              -- increase by i*0.250 cm
  props~setProperty("Height", box("int",newHeight)) -- set new Height
  text="oldHeight="oldHeight", newHeight="newHeight -- create info text
  xSheet~getCellByPosition(0,i)~setFormula(text) -- set cell to info text
end

::requires UNO.CLS                       -- get UNO support
```



Nutshell examples

Spreadsheet, Example 2/3

The screenshot shows the OpenOffice.org Calc application window titled "Untitled 1 - OpenOffice.org Calc". The interface includes a menu bar (File, Edit, View, Insert, Format, Tools, Data, Window, Help), a toolbar with various icons, and a formatting toolbar with options for font (Arial), size (10), bold (B), italic (I), underline (U), and alignment. The spreadsheet grid has columns A through H and rows 1 through 14. Row 1 contains the text "Hello, FOSDEM 2013!". Rows 2 through 6 show the text "oldHeight=432, newHeight=682", "oldHeight=432, newHeight=932", "oldHeight=432, newHeight=1182", and "oldHeight=432, newHeight=1432" respectively, indicating that the height of row 1 is being increased. Row 7 shows "oldHeight=432, newHeight=1682". The status bar at the bottom indicates "Sheet 1 / 3", "Default", "STD *", "Sum=0", and "100%".

	A	B	C	D	E	F	G	H
1	Hello, FOSDEM 2013!							
2	oldHeight=432, newHeight=682							
3	oldHeight=432, newHeight=932							
4	oldHeight=432, newHeight=1182							
5	oldHeight=432, newHeight=1432							
6	oldHeight=432, newHeight=1682							
7								
8								
9								
10								
11								
12								
13								
14								



Nutshell examples

Spreadsheet, Example 3/1

- Example 3
 - Create a spreadsheet document
 - Add text “**Hello, FOSDEM 2013!**” to A1
 - Demonstrate how to change the width of table columns



Nutshell examples

Spreadsheet, Example 3/2

```
xDesktop=uno.createDesktop()           -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader         -- get XComponentLoader interface

uri="private:factory/scalc"           -- new scalc document
doc=xcl~loadComponentFromURL(uri, "_blank", 0, .uno~noProps)

xSheets=doc~XSpreadSheetDocument~getSheets~XIndexAccess
xSheet =xSheets~getByIndex(0)~XSpreadSheet -- get first spreadsheet
                                           -- add entry to "A1"
xSheet~getCellByPosition(0,0)~setFormula("Hello, FOSDEM 2013!")

xCols=xSheet~XColumnRowRange~getColumns-- get XTableColumns

do i=1 to 5                            -- 0-based, hence columns # 2 (B) through # 6 (F)
  xCol=xCols~getByIndex(i)              -- fetch xCol
  props=xCol~XPropertySet                -- get access to its properties
  oldWidth=props~getPropertyValue("Width") -- get current value
  newWidth=oldWidth-i*250                -- decrease by i*0.250 cm
  props~setProperty("Width", box("int", newWidth)) -- set new Width
  text="oldWidth="oldWidth", newWidth="newWidth -- create info text
  xSheet~getCellByPosition(i,i)~setFormula(text) -- set cell to info text
end

::requires UNO.CLS                       -- get UNO support
```



Nutshell examples

Spreadsheet, Example 3/3

The screenshot shows the OpenOffice.org Calc application window titled "Untitled 1 - OpenOffice.org Calc". The interface includes a menu bar (File, Edit, View, Insert, Format, Tools, Data, Window, Help), a toolbar with various icons, and a status bar at the bottom. The spreadsheet grid has columns A through J and rows 1 through 23. Cell A1 contains the text "Hello, FOSDEM 2013!". The text in A1 is wrapped across several rows (2-6) in the adjacent columns (B-F). The wrapping is demonstrated by the text "oldWidth=2267, newWidth=2017" in row 2, "oldWidth=2267, newWidth=1767" in row 3, "oldWidth=2267, newWidth=1517" in row 4, "oldWidth=2267, newWidth=1267" in row 5, and "oldWidth=2267, newWidth=1017" in row 6. The status bar at the bottom shows "Sheet 1 / 3", "Default", "STD *", "Sum=0", and "100%".

	A	B	C	D	E	F	G	H	I	J
1	Hello, FOSDEM 2013!									
2		oldWidth=2267, newWidth=2017								
3		oldWidth=2267, newWidth=1767								
4		oldWidth=2267, newWidth=1517								
5		oldWidth=2267, newWidth=1267								
6		oldWidth=2267, newWidth=1017								
7										
8										
9										
10										
11										
12										
13										
14										
15										
16										
17										
18										
19										
20										
21										
22										
23										

Nutshell examples

Spreadsheet, Example 4/1

- Example 4
 - Create a spreadsheet document
 - Add text and a date
 - Demonstrate how to format individual cells and a cell range



Nutshell examples

Spreadsheet, Example 4/2

```
xDesktop=uno.createDesktop()           -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader          -- get XComponentLoader interface

uri="private:factory/scalc"            -- new scalc document
doc=xcl~loadComponentFromURL(uri,"_blank",0,.uno~noProps)

xSheets=doc~XSpreadSheetDocument~getSheets~XIndexAccess
xSheet =xSheets~getByIndex(0)~XSpreadSheet  -- get first spreadsheet

call uno.setCell xSheet, 0, 0, "Name:"    -- cell "A1"
call uno.setCell xSheet, "B1", "John Doe" -- cell "B1"
call uno.setCell xSheet, "A2", "Date:"    -- cell "A2"
call uno.setCell xSheet, 1, 1, "=TODAY()" -- cell "B2"
    -- format individual cells
xCellA2=xSheet~getCellByPosition(1, 0)    -- get access to cell "B1"
cbc=box("int", "CF E7 F5"x ~c2d)          -- define a RGB color
xCellA2~XPropertySet~setProperty("CellBackColor", cbc) -- set color

xCellB1=xSheet~getCellByPosition(1, 1)    -- get access to cell "B2"
cc=box("int", "c5 00 0b"x ~c2d)           -- define a RGB color
props=xCellB1~XPropertySet
props~setProperty("CharColor", cc)        -- set color
fontWeight=.uno_constants~new("com.sun.star.awt.FontWeight")
props~setProperty("CharWeight", fontWeight~semiBold)


    -- format using the properties of a XCellRange for "A1:A2"
props=xSheet~XCellRange~getCellRangeByName("A1:A2")~XPropertySet
props~setProperty("CharWeight", fontWeight~bold)

::requires UNO.CLS                      -- get UNO support
```



Nutshell examples

Spreadsheet, Example 4/3



The screenshot shows the OpenOffice.org Calc application window titled "Untitled 1 - OpenOffice.org Calc". The interface includes a menu bar (File, Edit, View, Insert, Format, Tools, Data, Window, Help), a toolbar with various icons, and a formatting toolbar. The spreadsheet grid is visible, with columns A through H and rows 1 through 23. The data entered is as follows:

	A	B	C	D	E	F	G	H
1	Name:	John Doe						
2	Date:	10/28/12						
3								
4								
5								
6								
7								
8								
9								
10								
11								
12								
13								
14								
15								
16								
17								
18								
19								
20								
21								
22								
23								

The status bar at the bottom shows "Sheet 1 / 3", "Default", "STD *", "Sum=0", and "100%".

Nutshell examples

Spreadsheet, Example 5/1

- Example 5
 - Create a spreadsheet document
 - Generate data for four quarters for 2011 and 2012
 - Format column headings
 - Format numbers
 - Create a chart from the generated data



Nutshell examples

Spreadsheet, Example 5/2a

```
xDesktop=uno.createDesktop()           -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader          -- get XComponentLoader interface

uri="private:factory/scalc"            -- new scalc document
doc=xcl~loadComponentFromURL(uri,"_blank",0,.uno~noProps)

xSheets=doc~XSpreadSheetDocument~getSheets~XIndexAccess
xSheet =xSheets~getByIndex(0)~XSpreadSheet  -- get first spreadsheet

call uno.setCell xSheet, "A1", "Quarter"
call uno.setCell xSheet, "B1", "2011"
call uno.setCell xSheet, "C1", "2012"
do i=1 to 4
    call uno.setCell xSheet, 0, i, "Q"i
    call uno.setCell xSheet, 1, i, random(0,5000)
    call uno.setCell xSheet, 2, i, random(0,5000)
end

props=xSheet~XCellRange~getCellRangeByName("A1:C1")~XPropertySet  -- column headings
fontWeight=.uno_constants~new("com.sun.star.awt.FontWeight")
props~setProperty("CharWeight", fontWeight~bold)

props=xSheet~XCellRange~getCellRangeByName("B2:C5")~XPropertySet  -- format numbers
props~setProperty("NumberFormat", 4) -- predefined style, format: "#,##0.00"

--> ... code to create a chart on next slide ...

::requires UNO.CLS           -- get UNO support
```



Nutshell examples

Spreadsheet, Example 5/2b

--> ... continued from previous slide: create a chart ...

```
structRect = .bsf~new("com.sun.star.awt.Rectangle") -- position & size of chart
structRect~X      = 300          -- x-offset: 0.300 cm
structRect~Y      = 2250         -- y-offset: 2.250 cm
structRect~Width  = 16000        -- width: 16.000 cm
structRect~Height = 8000         -- height: 8.000 cm

xRange=xSheet~XCellRange ~getCellRangeByName("A1:C5") -- data to be used for chart
rangeAddr = xRange~XCellRangeAddressable~getRangeAddress
arrAddr=bsf.createArrayOf(rangeAddr~getClass, rangeAddr) -- create array


xTableCharts = xSheet~XTableChartsSupplier~getCharts -- get Chart collection & insert
xTableCharts~addNewByName("FirstChart", structRect, arrAddr, .true, .true)

::requires UNO.CLS -- get UNO support
```



Nutshell examples

Spreadsheet, Example 5/3



Nutshell examples

Drawing (“sdraw”), 1

- 4 Services

DrawingDocument (com.sun.star.drawing.DrawingDocument),
DrawingDocumentFactory (com.sun.star.drawing.DrawingDocumentFactory),
GenericDrawingDocument (com.sun.star.drawing.GenericDrawingDocument),
OfficeDocument (com.sun.star.document.OfficeDocument)

- 20 Interfaces (unqualified)

XDocumentEventBroadcaster, XDocumentInfoSupplier,
XDocumentPropertiesSupplier, XDrawPageDuplicator, XDrawPagesSupplier,
XEmbeddedScripts, XEventBroadcaster, XEventsSupplier, XLayerSupplier,
XMasterPagesSupplier, XModel, XModifiable, XMultiServiceFactory,
XPrintJobBroadcaster, XPrintable, XPropertySet, XStorable,
XStyleFamiliesSupplier, XUndoManagerSupplier, XViewDataSupplier



Nutshell examples

Drawing (“sdraw”), 2

- 12 Properties

ApplyFormDesignMode, AutomaticControlFocus, BasicLibraries, BuildId, CharLocale, DialogLibraries, ForbiddenCharacters, HasValidSignatures, MapUnit, RuntimeUID, TabStop, VisibleArea



Nutshell examples

Drawing (“sdraw”), 3

- A collection of draw pages
- Each draw page
 - Allows any kind of drawing
 - Allows animation effects to be applied
- The draw concepts are fully reused for presentation documents!



Nutshell examples

Drawing, Example 1/1

- Example 1
 - Create a drawing document
 - Fetch the drawing component's service manager
 - Used to create shapes that can be stored with the document
 - Create and draw a rectangular shape, add it to the document
 - Set the shape's text to “**Hello, FOSDEM 2013!**”
 - Break up the text such that it fits into the rectangle



Nutshell examples

Drawing, Example 1/2

```
xDesktop=uno.createDesktop()           -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader          -- get XComponentLoader interface

uri="private:factory/sdraw"            -- new sdraw document
doc=xcl~loadComponentFromURL(uri,"_blank",0,.uno~noProps)

xsf=doc~XMultiServiceFactory          -- get the service manager (factory)
-- get access to the first draw page
xDrawPage = doc~XDrawPagesSupplier~getDrawPages~getByIndex(0)~XDrawPage

-- create a Rectangle shape and determine its position and size, add it to the page
xShape=xsf~createInstance("com.sun.star.drawing.RectangleShape") ~XShape
xShape~setPosition(.bsf~new("com.sun.star.awt.Point", 3000, 3000))
xShape~setSize(.bsf~new("com.sun.star.awt.Size", 5000, 2500))
xDrawPage~add(xShape)                  -- add new shape to first draw page


cr="0d"x                               -- ASCII carriage return char
xShape~XText~setString("Hello,"cr"FOSDEM"cr"2013!") -- now set string

::requires UNO.CLS                     -- get UNO support
```



Nutshell examples

Drawing, Example 1/3



Nutshell examples

Presentation (“simplpress”), 1

- 4 Services

DrawingDocumentFactory (com.sun.star.drawing.DrawingDocumentFactory),
GenericDrawingDocument (com.sun.star.drawing.GenericDrawingDocument),
OfficeDocument (com.sun.star.document.OfficeDocument),
PresentationDocument (com.sun.star.presentation.PresentationDocument)

- 23 Interfaces (unqualified)

XCustomPresentationSupplier, XDocumentEventBroadcaster,
XDocumentInfoSupplier, XDocumentPropertiesSupplier, XDrawPageDuplicator,
XDrawPagesSupplier, XEmbeddedScripts, XEventBroadcaster,
XEventsSupplier, XLayerSupplier, XLinkTargetSupplier, XMasterPagesSupplier,
XModel, XModifiable, XMultiServiceFactory, XPresentationSupplier,
XPrintJobBroadcaster, XPrintable, XPropertySet, XStorable,
XStyleFamiliesSupplier, XUndoManagerSupplier, XViewDataSupplier



Nutshell examples

Presentation (“simpres”), 2

- 12 Properties

ApplyFormDesignMode, AutomaticControlFocus, BasicLibraries, BuildId, CharLocale, DialogLibraries, ForbiddenCharacters, HasValidSignatures, MapUnit, RuntimeUID, TabStop, VisibleArea



Nutshell examples

Presentation (“simplpress”), 3

- A collection of draw pages
- Each draw page
 - Allows any kind of drawing
 - Allows animation effects to be applied
- Concept of “Master Pages”
 - Allows definition of specific layouts
- Layouts for title, listings, charts, etc.
- Presentation mode



Nutshell examples

Presentation, Example 1/1

- Example 1
 - Create a presentation document
 - Fetch its component's service manager
 - Used to create shapes that can be stored with the document
 - Create and draw a rectangular shape, add it to the document
 - Set the shape's text to “Hello, FOSDEM 2013!”
 - Break up the text such that it fits into the rectangle
 - Except for URL, the same code as for “sdraw”!



Nutshell examples

Presentation, Example 1/2

```
xDesktop=uno.createDesktop()           -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader         -- get XComponentLoader interface

uri="private:factory/simpres"         -- new simpres document
doc=xcl~loadComponentFromURL(uri, "_blank", 0, .uno~noProps)

xsf=doc~XMultiServiceFactory         -- get the service manager (factory)
-- get access to the first draw page
xDrawPage = doc~XDrawPagesSupplier~getDrawPages~getByIndex(0)~XDrawPage


-- create a Rectangle shape and determine its position and size
xShape=xsf~createInstance("com.sun.star.drawing.RectangleShape") ~XShape
xShape~setPosition(.bsf~new("com.sun.star.awt.Point", 3000, 3000))
xShape~setSize(.bsf~new("com.sun.star.awt.Size", 5000, 2500))

xDrawPage~add(xShape)                -- add new shape to first draw page
cr="0d"x                             -- ASCII carriage return char
xShape~XText~setString("Hello,"cr"FOSDEM"cr"2013!") -- now set string

::requires UNO.CLS                   -- get UNO support
```



Nutshell examples Presentation, Example 1/3



Nutshell examples

Presentation, Example 2/1

- Example 2
 - Create a presentation document
 - Create two pages with different layouts
 - One “Title Slide” page, layout number: 0
 - One “Title, Content” page, layout number: 1
 - Start the presentation at the end



Nutshell examples

Presentation, Example 2/2

```
xDesktop=uno.createDesktop()           -- bootstrap & get access to XDesktop
xcl=xDesktop~XComponentLoader         -- get XComponentLoader interface

uri="private:factory/simpres"         -- new simpres document
doc=xcl~loadComponentFromURL(uri,"_blank",0,.uno~noProps)

xDrawPages = doc~XDrawPagesSupplier~getDrawPages  -- get DrawPages

xDrawPage=xDrawPages~getByIndex(0)  -- get first (empty) page
xDrawPage~XPropertySet~setProperty("Layout", box("short",0))  -- "Title Slide"
xShapes=xDrawPage~XShapes           -- get access to its shapes
xShapes~getByIndex(0)~XText~setString("FOSDEM 2013")
xShapes~getByIndex(1)~XText~setString("Scripting Apache OpenOffice")

xDrawPage=xDrawPages~insertNewByIndex(1)~getByIndex(1)  -- insert at end, get access
xDrawPage~XPropertySet~setProperty("Layout", box("short",1))  -- "Title Content"
xShapes=xDrawPage~XShapes           -- get access to its shapes
xShapes~getByIndex(0)~XText~setString("Scripting Apache OpenOffice")

lf="0a"x                               -- define line-feed character
tab="09"x                               -- define tabulator character
str="First" lf"Second" lf tab "Second, 1" lf tab "Second, 2" lf"Third"
xShapes~getByIndex(1)~XText~setString(str)

doc~XPresentationSupplier~getPresentation~bsf.dispatch("start")  -- start presentation

::requires UNO.CLS                     -- get UNO support
```



Nutshell examples


Presentation, Example 2/3a

The screenshot displays the OpenOffice.org Impress application window titled "Untitled 1 - OpenOffice.org Impress". The interface includes a menu bar (File, Edit, View, Insert, Format, Tools, Slide Show, Window, Help) and a toolbar with various icons. The main workspace shows a slide with the title "FOSDEM 2013" and the subtitle "Scripting Apache OpenOffice". On the left, the "Slides" pane shows a list of slides, with the first slide selected. On the right, the "Tasks" pane is visible, showing a grid of layout templates. The "Title Slide" layout is highlighted with a mouse cursor and a callout box labeled "Title Slide". The status bar at the bottom indicates the current slide is "Slide 1 / 2" and the zoom level is "28%".



Nutshell examples

Presentation, Example 2/3b



Nutshell examples

Presentation, Example 3/1

- Example 3
 - Create a presentation document
 - Create two pages with different layouts
 - One “Title Slide” page, layout number: 0
 - One “Title, Content” page, layout number: 1
 - Use AOO's impress outline levels!
 - Kudos to Christoph Jopp, who found the property to use!
 - Start the presentation at the end



Nutshell examples

Presentation, Example 3/2

```
...
xText=xShapes~getByIndex(1)~XText    -- content's XText
call addItem xText, "First",          0 -- add string, determine level
call addItem xText, "Explored by many", 0
call addItem xText, "Kudos! go to",    1
call addItem xText, "Christoph Jopp!", 1
call addItem xText, "On 2012-11-07",  0, .false
...

::routine addItem                    -- adds string at the given (0-based outline) level
use arg xText, string, level, bNewParagraph=.true

xTR=xText~XTextRange~getEnd          -- get end, a XTextRange
xTR~XPropertySet~setProperty("NumberingLevel",level) -- set XTextRange level
xTR~setString(string)                -- set string

if bNewParagraph=.true then          -- add new paragraph
  xTR~getEnd~setString("\0a"x)       -- add linefeed character -> new paragraph

::routine dumpItems                  -- show level and string from XText
use arg xText

enum=xText~XEnumerationAccess~createEnumeration -- enumerate paragraphs
do i=1 while enum~hasMoreElements
  xtr=enum~nextElement~XTextRange -- we need XTextRange's string & properties

  nl=xtr~XPropertySet~getPropertyValue("NumberingLevel")
  say "    item #" i": NumberingLevel="pp(nl) pp(xtr~getString)
end
```



Nutshell examples Presentation, Example 2/3

The screenshot shows the OpenOffice Impress application window. The title bar reads "Untitled 1 - OpenOffice.org Impress". The main window displays a presentation slide with the following content:

- First
- Explored by many
 - Kudos! go to
 - Christoph Jopp!
- On 2012-11-07

The slide is titled "Scripting Apache OpenOffice". The left sidebar shows a "Slides" panel with two slides: "1 ApacheCon Europe 2012" and "2 Scripting Apache OpenOffice". The right sidebar shows a "Tasks" panel with options like "Master Pages", "Layouts", "Table Design", and "Custom Animation". The bottom status bar indicates "Slide 2 / 2 Default" and "67%".



Nutshell examples

URE (UNO Runtime Environment)

- There are UNO types that can be used independently of the AOO GUI! E.g.
 - `"com.sun.star.lang.Locale"`
 - `"com.sun.star.linguistic2.LinguServiceManager"`
- Can therefore be used by/incorporated into any other application!
- Need to bootstrap and connect to the UNO runtime environment (URE)
 - Fetch its service manager
 - Instantiate services
 - Use services, request their interfaces



Nutshell examples

URE, Spellchecker Example 1/1

- Example “Spellchecker”
 - Create a connection to URE
 - Get its service manager
 - Used to create the spellchecker service via `"com.sun.star.linguistic2.LinguServiceManager"`
 - Use all locales available to the spellchecker
 - In this example: some English locales
 - Spellcheck the word “thru” with the different English locales
 - If not correct, list the alternatives of the locale



Nutshell examples

URE, Spell Checker Example 1/2

```
xContext = UNO.connect()           -- bootstrap and connect to URE
xSM = xContext~getServiceManager   -- get the service manager

serviceName="com.sun.star.linguistic2.LinguServiceManager"
lsm=xsm~createInstanceWithContext(serviceName, xContext) -- create the service
xSpellChecker = lsm~XLinguServiceManager~getSpellChecker -- get the spell checker
locales=xSpellChecker~XSupportedLocales~getLocales      -- get all supported locales

word="thru" -- word to spellcheck
do locale over locales -- iterate over all available Locales
  str=locale~language/"locale~country"/"locale~variant "-> word:" pp(word)":"
  ok=xSpellChecker~isValid(word, locale, .UNO~noProps) -- check word
  if ok then str=str "correct"
    else str=str "NOT correct! Available alternatives:"
  say str

  if \ok then -- not correct, get & show alternatives
  do
    alternatives=xSpellChecker~spell(word, locale, .UNO~noProps)
    if alternatives <> .nil then
    do
      do a over alternatives~getAlternatives
        say "0909"x pp(a)
      end
    end
  end
end
end

::requires UNO.CLS -- get UNO support
```



Nutshell examples

URE, Spell Checker Example 1/3

```
E:\2013\FOSDEM\vortrag\code>rexx spellcheck1.rxo
en/US/ -> word: [thru]: correct
en/GB/ -> word: [thru]: NOT correct! Available alternatives:
        [thrum]
        [thou]
        [thrush]
        [thrust]
        [Thur]
        [truth]
        [three]
        [threw]
en/AU/ -> word: [thru]: NOT correct! Available alternatives:
        [threw]
        [thro]
        [through]
        [thrum]
        [thou]
en/CA/ -> word: [thru]: correct
en/NZ/ -> word: [thru]: NOT correct! Available alternatives:
        [through]
        [thrum]
        [thou]
en/ZA/ -> word: [thru]: NOT correct! Available alternatives:
        [thrum]
        [thou]
        [thrush]
        [thrust]
        [Thur]
        [truth]
        [through]
        [three]
```



Roundup

- UNO
- Very Powerful
 - Complex
 - Documentation, examples very important
- Creating, editing AOO documents
 - swriter, scalc, sdraw, simpres
- URE
- Need for many more nutshell examples in all programming languages!



Links to ooRexx/BSF4ooRexx

- ooRexx (as of 2013-01-31, version: 4.1.2)
 - An easy to learn and easy to use scripting language
 - Compatible to (“classic”) Rexx
 - Developed originally by IBM (“Object REXX”)
 - Source code was received by the non-for-profit SIG “Rexx Language Association (<http://www.RexxLA.org>)”
 - Opensourced as “Open Object Rexx (ooRexx)”
 - Home: <http://www.ooRexx.org>
 - Downloads: <http://sourceforge.net/projects/oorex/ files/oorex/>
 - Brief overview (since opensourcing a lot got added):
http://wi.wu.ac.at/rgf/rexx/misc/ecoop06/ECOOP2006_RDL_Workshop_Flatscher_Paper.pdf
 - Authoring a new book that introduces ooRexx



Links to ooRexx/BSF4ooRexx

- **BSF4ooRexx (with built-in AOO/LO support)**
 - Allows to use all of Java from ooRexx as if it was an interpreted, typeless and caseless language!
 - “Camouflaging Java as ooRexx” (package “**BSF.CLS**”)
 - All Java classes and Java objects look like ooRexx' ones!
 - Includes specific AOO support (package “**UNO.CLS**”)
 - Developed since 2000 to allow the creation of platform independent Rexx and ooRexx scripts
 - Using Apache's “Bean Scripting Framework (BSF)”, cf. <http://commons.apache.org/bsf/>
 - Home: <http://sourceforge.net/projects/bsf4oorex/>
 - Downloads: <http://sourceforge.net/projects/bsf4oorex/files/GA/>

